
Technical Reference (excerpt)

Credit Operations and Partner Sales semantic model
How the Power BI model behind the Operations Workflow Dashboard and the Referral Partner
Sales Dashboard is built, so Larkspur's IT staff can maintain it after handover.

Sample data. All names, organizations, and figures are fictional.

1. Purpose and scope
One semantic model serves two reports. The Operations Workflow Dashboard is used by the credit manager to watch
volume, turn time, document requests, decisions, and reversals. The Referral Partner Sales Dashboard is used by the
sales manager and the ten partner sales reps to track submissions, funding conversion, and the health of each
referral partner. This excerpt covers the data model, tables, relationships, calculated objects, the main measures,
and refresh notes. The full reference also lists every visual by page and the row-level security plan.

2. Data model
Applications is the central fact table, with one row per financing application. Workflow Actions and Document
Requests hold the steps and document requests for each application and are filtered through Applications, so any
slicer that affects applications also affects them. Rep filters reach applications through Partners, because each
partner is covered by one rep.

Active relationship Inactive (used with USERELATIONSHIP)
Not related to other tables: Flow Stage, Pipeline Stage, Score Bins, Score Threshold

Figure 1. Model view. Only key columns are shown. Each line runs from the many side (*) to the one side (1).

One relationship is inactive on purpose. FundedDate lets funded measures count deals by the month they funded,
while everything else counts by the month received. The Calendar table comes from the source rather than from DAX,
because it carries the company's holidays and the number of business days left in each month, which the month-end
turn time measures depend on.

3. Tables

Table Type Rows Grain Key Source

Applications Fact 5,973 One row per financing application ApplicationID Applications
sheet of the
export

Workflow Actions Fact 47,276 One row per workflow step on an
application

ActionID Workflow Actions
sheet

Document
Requests

Fact 9,625 One row per document requested
from an applicant

RequestID Document
Requests sheet

Officers Dimension 13 One row per credit officer OfficerID Officers sheet

Channels Dimension 5 One row per application source ChannelID Channels sheet

Equipment
Categories

Dimension 7 One row per equipment category EquipmentCategory Equipment
Categories sheet

Partners Dimension 404 One row per referral partner PartnerID Partners sheet

Sales Reps Dimension 10 One row per partner sales rep RepID Sales Reps sheet

States Dimension 18 One row per state served StateCode States sheet

Calendar Dimension 640 One row per day, with business-day
flags and business days left in the
month

Date Calendar sheet

Flow Stage Calculated 6 Steps for the application flow funnel Stage DAX calculated
table

Pipeline Stage Calculated 5 Steps for the partner pipeline funnel Stage DAX calculated
table

Score Bins Calculated 20 Five-point bins for the health score
chart

Bin DAX calculated
table

Score Threshold Parameter 4 Threshold values for the partner
health list (35 to 70)

Score Threshold DAX calculated
table

4. Relationships

From (many side) To (one side) Active Purpose

Applications[ReceivedDate] Calendar[Date] Active Filters by the date the
application was received

Applications[FundedDate] Calendar[Date] Inactive Used by funded measures
through USERELATIONSHIP

From (many side) To (one side) Active Purpose

Applications[ChannelID] Channels[ChannelID] Active

Applications[EquipmentCategory] 'Equipment
Categories'[EquipmentCategory]

Active

Applications[OfficerID] Officers[OfficerID] Active The credit officer assigned to
the application

Applications[PartnerID] Partners[PartnerID] Active Blank for direct and
marketplace applications

Partners[RepID] 'Sales Reps'[RepID] Active Rep filters reach applications
through Partners

Applications[StateCode] States[StateCode] Active Location of the borrowing
business

'Workflow
Actions'[ApplicationID]

Applications[ApplicationID] Active Actions inherit every
application filter

'Document
Requests'[ApplicationID]

Applications[ApplicationID] Active Requests inherit every
application filter

Every relationship is many-to-one with a single cross-filter direction. No relationship is bidirectional, which keeps
filter paths predictable.

5. Calculated tables and columns
The two stage tables drive the funnel visuals, Score Bins drives the health score chart, and Score Threshold feeds the
partner list slicer. The calculated column below finds how many business days were left in the month when an
application's file became complete.

File Complete Days Left =
LOOKUPVALUE ( Calendar[BusinessDaysLeftInMonth], Calendar[Date], Applications[File Complete Date] )

Calculated column Purpose

Applications[File Complete Date] Date the last requested document arrived

Applications[File Complete Days Left] Business days left in the month when the file was complete

Applications[File Complete Window] Month-end window flag used on the Turn time page

Applications[Days Left Bucket] Axis for the decision wait chart, sorted by Days Left Sort

Applications[Days Left Sort] Sort column for Days Left Bucket

Applications[Exit Point] Where applications that did not fund left the process

Applications[Outcome Group] Outcome group for the sales pages

Channels[Channel Comparison] Legend for the conversion trend line

Equipment Categories[Category Comparison] Legend for the document wait trend line

6. Measure catalog
The model holds 76 measures in a table named _Measures, grouped into display folders. The 20 below carry most of
the logic; the rest are ratios or variations built from them. Turn time and waits are stored in business days, so the
measures do not convert units.

Applications
Core, whole number

Count of applications in the current filter context.

COUNTROWS ( Applications )

Approval Rate
Core, percentage, 1 decimal

Approved share of decided applications.

DIVIDE ( [Approved Applications], [Decided Applications] 
)

Funding Conversion
Core, percentage, 1 decimal

Share of applications received that went on to fund.

DIVIDE ( [Funded Applications], [Applications] )

Funded Volume
Core, currency, no decimals

Dollars funded in the selected months, by funding date.

CALCULATE (
    SUM ( Applications[FundedAmount] ),
    USERELATIONSHIP ( Calendar[Date], 
Applications[FundedDate] )
)

Median Turn Time
Turn Time, decimal, 1 place

Median business days from receipt to the credit decision. MEDIAN
skips undecided applications because their value is blank.

MEDIAN ( Applications[TurnTimeBusinessDays] )

Month-End Decision Wait
Turn Time, decimal, 1 place

Median decision wait for files completed in the last three business
days of a month.

CALCULATE (
    [Median Decision Wait],
    Applications[File Complete Window] = "Last 3 
business days"
)

Median Document Wait
Documents, decimal, 1 place

Median business days from a request to receipt of the document.
Outstanding requests are blank and skipped.

MEDIAN ( 'Document Requests'[WaitBusinessDays] )

Approvals Reversed
Reversals, whole number

Approvals reversed before funding.

CALCULATE ( [Applications], Applications[Status] = 
"Reversed" )

Reversal Rate
Reversals, percentage, 1 decimal

Share of approvals that were later reversed.

DIVIDE ( [Approvals Reversed], [Approved Applications] )

Highest Reversal Officer
Reversals, text

Officer with the highest reversal rate among officers with at least 60
approvals.

VAR t =
    TOPN (
        1,
        FILTER ( ALL ( Officers[OfficerName] ), 
[Approved Applications] >= 60 ),
        [Reversal Rate]
    )
RETURN
    MAXX ( t, Officers[OfficerName] )

Same Level Average Decisions
Comparisons, whole number

Average decisions per officer at the selected officer's level.

VAR lvl = SELECTEDVALUE ( Officers[OfficerLevel] )
RETURN
    AVERAGEX ( FILTER ( ALL ( Officers ), 
Officers[OfficerLevel] = lvl ), [Decided Applications] )

Flow Count
Application Flow, whole number

Value for each step of the funnel. Put 'Flow Stage'[Stage] on
Category and this measure on Values.

SWITCH (
    SELECTEDVALUE ( 'Flow Stage'[Stage] ),
    "Received", [Applications],
    "Decided", [Decided Applications],
    "Approved", [Approved Applications],
    "Contract sent", [Contracts Sent],
    "Contract signed", [Contracts Signed],
    "Funded", [Funded Applications]
)

Contracts Out Not Signed
Partner Sales, currency, no decimals

Requested amount on approved deals with a contract out and no
signature yet.

VAR sentApps =
    CALCULATETABLE (
        VALUES ( 'Workflow Actions'[ApplicationID] ),
        'Workflow Actions'[ActionType] = "Contract 
sent",
        REMOVEFILTERS ( Calendar )
    )
RETURN
    CALCULATE (
        SUM ( Applications[RequestedAmount] ),
        REMOVEFILTERS ( Calendar ),
        TREATAS ( sentApps, Applications[ApplicationID] 
),
        Applications[ContractSigned] = "No",
        Applications[Status] = "In progress"
    )

Activity Points
Partner Health, decimal, 0 places

Up to 40 points for keeping the usual pace.

VAR u = [Usual Pace]
VAR r = [Recent Pace]
VAR ratio = IF ( u > 0, DIVIDE ( r, u ), IF ( r > 0, 1, 
0 ) )
RETURN
    40 * MIN ( 1, ratio )

Conversion Points
Partner Health, decimal, 0 places

Up to 30 points. A 60% funding conversion earns all 30.

30 * MIN ( 1, [Conversion 12 Months] / 0.6 )

Recency Points
Partner Health, decimal, 0 places

Up to 20 points, falling to zero at 90 days without a submission.

VAR ds = [Days Since Last Submission]
RETURN
    SWITCH ( TRUE (), ds <= 14, 20, ds >= 90, 0, 20 * ( 
90 - ds ) / 76 )

Is Scored
Partner Health, whole number

1 when a partner has at least six months of history and three
submissions in the last year.

VAR asOf = [As Of Date]
VAR onb = SELECTEDVALUE ( Partners[OnboardDate] )
VAR n12 =
    CALCULATE ( [Applications], REMOVEFILTERS ( Calendar 
), Applications[ReceivedDate] > asOf - 365 )
RETURN
    IF ( onb <= asOf - 180 && n12 >= 3, 1, 0 )

Health Score
Partner Health, whole number

Partner health score from 0 to 100. Blank for partners without
enough history.

IF (
    [Is Scored] = 1,
    ROUND ( [Activity Points] + [Conversion Points] + 
[Recency Points] + [Approval Points], 0 )
)

Health Band
Partner Health, text

At risk under 45, Watch from 45 to 69, Healthy at 70 and up.

VAR s = [Health Score]
RETURN
    IF ( ISBLANK ( s ), BLANK (), IF ( s < 45, "At 
risk", IF ( s < 70, "Watch", "Healthy" ) ) )

Main Issue
Partner Health, text

The score part with the biggest shortfall, written as a short phrase
for the list and the outreach sheet.

VAR ga = 40 - [Activity Points]
VAR gc = 30 - [Conversion Points]
VAR gr = 20 - [Recency Points]
VAR gp = 10 - [Approval Points]
VAR mx = MAX ( MAX ( ga, gc ), MAX ( gr, gp ) )
VAR u = [Usual Pace]
RETURN
    SWITCH (
        TRUE (),
        mx <= 0.5, "No weak area",
        mx = ga,
            IF ( [Recent Pace] = 0, "No submissions in 
90 days",
                "Activity down " & FORMAT ( 1 - DIVIDE ( 
[Recent Pace], u ), "0%" ) ),
        mx = gc, "Funding conversion " & FORMAT ( 
[Conversion 12 Months], "0%" ),
        mx = gr, "Last submission " & [Days Since Last 
Submission] & " days ago",
        "Approval rate " & FORMAT ( [Approval 12 
Months], "0%" )
    )

7. Refresh notes
Source. One Excel workbook, Larkspur Data Tables.xlsx, produced by the loan origination system's nightly export
and saved to the Reporting folder of the Credit Operations SharePoint site. Each of the ten tables is an Excel table,

and Power Query reads the tables rather than the sheets so extra rows or notes on a sheet do not load.

Schedule. The semantic model refreshes daily at 6:30 AM Eastern, after the export lands at about 5:00 AM.
Because the workbook lives in SharePoint Online, Power BI connects to it directly and no on-premises data
gateway is needed.

Credentials. The data source uses an organizational account owned by the reporting service account, not a
person, so refresh keeps working when staff change.

Column protection. Each query keeps only the expected columns with Table.SelectColumns  and
MissingField.UseNull , so a dropped column loads as empty instead of failing the refresh. The Data Refresh
Guide covers what to do next.

Business days. Turn time, decision wait, and document wait arrive already calculated in business days (8:00 AM
to 6:00 PM, Monday through Friday, excluding company holidays). The model does not recalculate them, so a
change to the holiday calendar belongs in the export and its Calendar sheet.

As-of date. The partner health measures count back from the latest application date in the data, not from today's
date, so scores stay correct even if a refresh is missed for a day.

Checks. After each refresh, the About sheet of the workbook lists row counts and totals that should match the
report with all slicers cleared (5,973 applications, 47,276 workflow actions, and 9,625 document requests in this
sample).

